﻿/**
 * @author sminutoli
 * versión 1.0
 * Botón que dispara un tooltip después de x cantidad de tiempo;
 */
import com.dinmax.flash.draw.cabeza.ToolTipCabeza

class com.dinmax.flash.draw.cabeza.ButtonCabezaTooltip
extends com.dinmax.flash.draw.cabeza.ButtonCabeza {
	
	private var sToolTipMsg:String;
	private var nToolTipTimer:Number;
	private var mcTimer:MovieClip;
	private var tToolTip:ToolTipCabeza;
	
	function onRollOver():Void{
		super.onRollOver();
		var root:ButtonCabezaTooltip = this;
		var nTimer:Number= getTimer();
		mcTimer= this.createEmptyMovieClip("mcTimer", this.getNextHighestDepth());
		
		mcTimer.onEnterFrame=function(){
			var nDif:Number= getTimer() - nTimer;
			
			if(nDif > root.getToolTipTimer()){
				//trace("ButtonCabezaTooltip.as | mcTimer.onEnterFrame() | tiempo excedido");
				delete this.onEnterFrame;
				root.tToolTip = ToolTipCabeza(root.attachMovie("ToolTip", "mcToolTip", root.getNextHighestDepth()));
				//trace("ButtonCabezaTooltip.as | toolTip= "+ root.tToolTip);
				root.tToolTip.setTooltipMsg(root.sToolTipMsg);
				root.tToolTip._x = _xmouse;
				root.tToolTip._y = _ymouse + root.tToolTip._height;
				var toolPos:MovieClip = root.createEmptyMovieClip("toolPos", root.getNextHighestDepth());
				toolPos.onEnterFrame= function ():Void{
					root.tToolTip._x = _xmouse;
					root.tToolTip._y = _ymouse + root.tToolTip._height;
				}
			};
			
		};
		
	};
	
	function onRollOut():Void{
		super.onRollOut();
		//this.mcBorde._visible=false;
		delete mcTimer.onEnterFrame;
		this.tToolTip.remove();
	};
	
	function onRelease():Void {
		super.onRelease();
		delete mcTimer.onEnterFrame;
		this.tToolTip.remove();
	};
		
	public function getToolTipMsg():String {
		return sToolTipMsg;
	}

	public function setToolTipMsg(sToolTipMsg:String):Void {
		this.sToolTipMsg = sToolTipMsg;
	}

	public function getToolTipTimer():Number {
		return nToolTipTimer;
	}

	public function setToolTipTimer(nToolTipTimer:Number):Void {
		//trace("ButtonCabezaTooltip.as | "+nToolTipTimer);
		this.nToolTipTimer = nToolTipTimer;
	}

	public function getToolTip():ToolTipCabeza {
		return tToolTip;
	}

	public function setToolTip(tToolTip:ToolTipCabeza):Void {
		this.tToolTip = tToolTip;
	}

}